

> **restart:**

> **with(linalg):**

Warning, new definition for norm

Warning, new definition for trace

Viga de Bernoulli-Euler

Jose M^a Goicolea Ruigomez,
10/02/2000

Funciones de forma cúbicas del elemento

> **N:=matrix([[1-3*(x/l)^2+2*(x/l)^3,x*(1-2*x/l+(x/l)^2),
(x/l)^2*(3-2*x/l),x^2/l*(x/l-1)]]);**

$$N := \begin{bmatrix} 1 - 3\frac{x^2}{l^2} + 2\frac{x^3}{l^3} & x \left(1 - 2\frac{x}{l} + \frac{x^2}{l^2} \right) & \frac{x^2 \left(3 - 2\frac{x}{l} \right)}{l^2} & \frac{x^2 \left(\frac{x}{l} - 1 \right)}{l} \end{bmatrix}$$

Vector de grados de libertad del elemento: flechas y derivadas de las mismas (=giros)

> **a:=matrix([[w[1]],[wx[1]],[w[2]],[wx[2]]]);**

$$\alpha := \begin{bmatrix} w_1 \\ wx_1 \\ w_2 \\ wx_2 \end{bmatrix}$$

Interpolación de flechas

> **w[h](x):=multiply(N,a)[1,1];**

$$w_h(x) := \left(1 - 3\frac{x^2}{l^2} + 2\frac{x^3}{l^3} \right) w_1 + x \left(1 - 2\frac{x}{l} + \frac{x^2}{l^2} \right) wx_1 + \frac{x^2 \left(3 - 2\frac{x}{l} \right) w_2}{l^2} + \frac{x^2 \left(\frac{x}{l} - 1 \right) wx_2}{l}$$

Interpolación de curvaturas

> **kappa[h](x):=diff(w[h](x),x,x);**

$$\kappa_h(x) := \left(-6 \frac{1}{l^2} + 12 \frac{x}{l^3} \right) w_1 + 2 \left(-2 \frac{1}{l} + 2 \frac{x}{l^2} \right) w_{x1} + 2 \frac{x w_{x1}}{l^2} + 2 \frac{\left(3 - 2 \frac{x}{l} \right) w_2}{l^2} - 8 \frac{x w_2}{l^3} + 2 \frac{\left(\frac{x}{l} - 1 \right) w_{x2}}{l} + 4$$

Matriz de interpolación de deformaciones (curvaturas)

> **B:=map(z->diff(z,x,x),N);**

$$B := \begin{bmatrix} -6 \frac{1}{l^2} + 12 \frac{x}{l^3} & -4 \frac{1}{l} + 6 \frac{x}{l^2} & 2 \frac{3 - 2 \frac{x}{l}}{l^2} - 8 \frac{x}{l^3} & 2 \frac{\frac{x}{l} - 1}{l} + 4 \frac{x}{l^2} \end{bmatrix}$$

La curvatura (aproximada) es

> **kappa[h](x):=multiply(B,a)[1,1];**

$$\kappa_h(x) := \left(-6 \frac{1}{l^2} + 12 \frac{x}{l^3} \right) w_1 + \left(-4 \frac{1}{l} + 6 \frac{x}{l^2} \right) w_{x1} + \left(2 \frac{3 - 2 \frac{x}{l}}{l^2} - 8 \frac{x}{l^3} \right) w_2 + \left(2 \frac{\frac{x}{l} - 1}{l} + 4 \frac{x}{l^2} \right) w_{x2}$$

Integrando de la matriz de rigidez del elemento

> **Kb:=multiply(transpose(B),(E*J)*B);**

$Kb :=$

$$\begin{bmatrix} EJ \left(-6 \frac{1}{l^2} + 12 \frac{x}{l^3} \right)^2, \left(-6 \frac{1}{l^2} + 12 \frac{x}{l^3} \right) EJ \left(-4 \frac{1}{l} + 6 \frac{x}{l^2} \right), \left(-6 \frac{1}{l^2} + 12 \frac{x}{l^3} \right) EJ \left(2 \frac{3 - 2 \frac{x}{l}}{l^2} - 8 \frac{x}{l^3} \right), \\ \left(-6 \frac{1}{l^2} + 12 \frac{x}{l^3} \right) EJ \left(2 \frac{\frac{x}{l} - 1}{l} + 4 \frac{x}{l^2} \right) \\ \left(-4 \frac{1}{l} + 6 \frac{x}{l^2} \right) EJ \left(-4 \frac{1}{l} + 6 \frac{x}{l^2} \right), EJ \left(-4 \frac{1}{l} + 6 \frac{x}{l^2} \right)^2, \left(-4 \frac{1}{l} + 6 \frac{x}{l^2} \right) EJ \left(2 \frac{3 - 2 \frac{x}{l}}{l^2} - 8 \frac{x}{l^3} \right), \\ \left(-4 \frac{1}{l} + 6 \frac{x}{l^2} \right) EJ \left(2 \frac{\frac{x}{l} - 1}{l} + 4 \frac{x}{l^2} \right) \\ \left(-6 \frac{1}{l^2} + 12 \frac{x}{l^3} \right) EJ \left(2 \frac{3 - 2 \frac{x}{l}}{l^2} - 8 \frac{x}{l^3} \right), \left(-4 \frac{1}{l} + 6 \frac{x}{l^2} \right) EJ \left(2 \frac{3 - 2 \frac{x}{l}}{l^2} - 8 \frac{x}{l^3} \right), \left(2 \frac{3 - 2 \frac{x}{l}}{l^2} - 8 \frac{x}{l^3} \right)^2 EJ, \\ \left(-6 \frac{1}{l^2} + 12 \frac{x}{l^3} \right) EJ \left(2 \frac{\frac{x}{l} - 1}{l} + 4 \frac{x}{l^2} \right), \left(-4 \frac{1}{l} + 6 \frac{x}{l^2} \right) EJ \left(2 \frac{\frac{x}{l} - 1}{l} + 4 \frac{x}{l^2} \right) \end{bmatrix}$$

$$\left[\begin{array}{c} \left(\begin{array}{c} 3 - 2\frac{x}{l} \\ 2\frac{x}{l^2} - 8\frac{x}{l^3} \end{array} \right) EJ \left(\begin{array}{c} \frac{x}{l} - 1 \\ 2\frac{x}{l} + 4\frac{x}{l^2} \end{array} \right) \\ \left(\begin{array}{c} -6\frac{1}{l^2} + 12\frac{x}{l^3} \end{array} \right) EJ \left(\begin{array}{c} \frac{x}{l} - 1 \\ 2\frac{x}{l} + 4\frac{x}{l^2} \end{array} \right), \left(\begin{array}{c} -4\frac{1}{l} + 6\frac{x}{l^2} \end{array} \right) EJ \left(\begin{array}{c} \frac{x}{l} - 1 \\ 2\frac{x}{l} + 4\frac{x}{l^2} \end{array} \right), \left(\begin{array}{c} 3 - 2\frac{x}{l} - 8\frac{x}{l^3} \end{array} \right) EJ \left(\begin{array}{c} \frac{x}{l} - 1 \\ 2\frac{x}{l} \end{array} \right) \\ EJ \left(\begin{array}{c} \frac{x}{l} - 1 \\ 2\frac{x}{l} + 4\frac{x}{l^2} \end{array} \right)^2 \end{array} \right]$$

Integración para obtener la matriz de rigidez

> **Kb:=map(z->int(z,x=0..l),Kb);**

$$Kb := \begin{bmatrix} 12\frac{EJ}{l^3} & 6\frac{EJ}{l^2} & -12\frac{EJ}{l^3} & 6\frac{EJ}{l^2} \\ 6\frac{EJ}{l^2} & 4\frac{EJ}{l} & -6\frac{EJ}{l^2} & 2\frac{EJ}{l} \\ -12\frac{EJ}{l^3} & -6\frac{EJ}{l^2} & 12\frac{EJ}{l^3} & -6\frac{EJ}{l^2} \\ 6\frac{EJ}{l^2} & 2\frac{EJ}{l} & -6\frac{EJ}{l^2} & 4\frac{EJ}{l} \end{bmatrix}$$

Viga de Timoshenko

Vector de grados de libertad: flechas y giros independientes

> **a:=matrix([[w[1]],[theta[1]],[w[2]],[theta[2]]]);**

$$a := \begin{bmatrix} w_1 \\ \theta_1 \\ w_2 \\ \theta_2 \end{bmatrix}$$

Interpolacion de las flechas (lineal)

> **Nx:=matrix([[1-x/l,0,x/l,0]]);**

$$Nx := \begin{bmatrix} 1 - \frac{x}{l} & 0 & \frac{x}{l} & 0 \end{bmatrix}$$

Interpolacion de los giros (lineal)

> **Nt:=matrix([[0,1-x/l,0,x/l]]);**

$$Nt := \begin{bmatrix} 0 & 1 - \frac{x}{l} & 0 & \frac{x}{l} \end{bmatrix}$$

Flechas y giros interpolados

> **w[h](x):=multiply(Nx,a)[1,1];**

> **theta[h](x):=multiply(Nt,a)[1,1];**

$$w_h(x) := \left(1 - \frac{x}{l}\right) w_1 + \frac{x}{l} w_2$$

$$\theta_h(x) := \left(1 - \frac{x}{l}\right) \theta_1 + \frac{x}{l} \theta_2$$

Curvatura aproximada

> $\kappa_h(x) := \frac{\partial}{\partial x} \theta_h(x)$

$$\kappa_h(x) := -\frac{\theta_1}{l} + \frac{\theta_2}{l}$$

Cortante aproximado

> $\alpha_h(x) := \left(\frac{\partial}{\partial x} w_h(x)\right) - \theta_h(x)$

$$\alpha_h(x) := -\frac{w_1}{l} + \frac{w_2}{l} - \left(1 - \frac{x}{l}\right) \theta_1 - \frac{x}{l} \theta_2$$

Funciones de interpolacion de deformaciones, para flexion y cortante

> $Bf := \text{map}\left(z \rightarrow \frac{\partial}{\partial x} z, Nt\right)$

$$Bc := \text{evalm}\left(\text{map}\left(z \rightarrow \frac{\partial}{\partial x} z, Nx\right) - Nt\right)$$

$$Bf := \begin{bmatrix} 0 & -\frac{1}{l} & 0 & \frac{1}{l} \end{bmatrix}$$

$$Bc := \begin{bmatrix} -\frac{1}{l} & \frac{x}{l} - 1 & \frac{1}{l} & -\frac{x}{l} \end{bmatrix}$$

Las curvaturas y cortantes aproximados salen lo mismo que antes:

$$\text{> } \mathbf{kappa[h](x) := \text{multiply}(Bf, a)[1,1];$$

$$\kappa_h(x) := -\frac{\theta_1}{l} + \frac{\theta_2}{l}$$

$$\text{> } \mathbf{alfa[h](x) := \text{multiply}(Bc, a)[1,1];$$

$$\alpha_h(x) := -\frac{w_1}{l} + \left(\frac{x}{l} - 1\right) \theta_1 + \frac{w_2}{l} - \frac{x \theta_2}{l}$$

Matriz de rigidez por flexión

$$\text{> } \mathbf{Kf := \text{multiply}(\text{transpose}(Bf), (E * J) * Bf);}$$

$$Kf := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & \frac{EJ}{l^2} & 0 & -\frac{EJ}{l^2} \\ 0 & 0 & 0 & 0 \\ 0 & -\frac{EJ}{l^2} & 0 & \frac{EJ}{l^2} \end{bmatrix}$$

$$\text{> } \mathbf{Kf := \text{map}(z \rightarrow \text{int}(z, x=0..l), Kf);}$$

$$Kf := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & \frac{EJ}{l} & 0 & -\frac{EJ}{l} \\ 0 & 0 & 0 & 0 \\ 0 & -\frac{EJ}{l} & 0 & \frac{EJ}{l} \end{bmatrix}$$

Matriz de rigidez por cortante

> **Kc:=multiply(transpose(Bc),(G*A)*Bc);**

$$Kc := \begin{bmatrix} \frac{GA}{l^2} & -\frac{GA\left(\frac{x}{l}-1\right)}{l} & -\frac{GA}{l^2} & \frac{GAx}{l^2} \\ -\frac{GA\left(\frac{x}{l}-1\right)}{l} & GA\left(\frac{x}{l}-1\right)^2 & \frac{GA\left(\frac{x}{l}-1\right)}{l} & -\frac{\left(\frac{x}{l}-1\right)GAx}{l} \\ -\frac{GA}{l^2} & \frac{GA\left(\frac{x}{l}-1\right)}{l} & \frac{GA}{l^2} & -\frac{GAx}{l^2} \\ \frac{GAx}{l^2} & -\frac{\left(\frac{x}{l}-1\right)GAx}{l} & -\frac{GAx}{l^2} & \frac{x^2 GA}{l^2} \end{bmatrix}$$

> **Kc:=map(z->int(z,x=0..l),Kc);**

$$Kc := \begin{bmatrix} \frac{GA}{l} & \frac{1}{2}GA & -\frac{GA}{l} & \frac{1}{2}GA \\ \frac{1}{2}GA & \frac{1}{3}lGA & -\frac{1}{2}GA & \frac{1}{6}lGA \\ -\frac{GA}{l} & -\frac{1}{2}GA & \frac{GA}{l} & -\frac{1}{2}GA \\ \frac{1}{2}GA & \frac{1}{6}lGA & -\frac{1}{2}GA & \frac{1}{3}lGA \end{bmatrix}$$

Matriz de rigidez conjunta

> **Kt:=evalm(Kf+Kc);**

$$Kt := \begin{bmatrix} \frac{GA}{l} & \frac{1}{2}GA & -\frac{GA}{l} & \frac{1}{2}GA \\ \frac{1}{2}GA & \frac{EJ}{l} + \frac{1}{3}lGA & -\frac{1}{2}GA & -\frac{EJ}{l} + \frac{1}{6}lGA \\ -\frac{GA}{l} & -\frac{1}{2}GA & \frac{GA}{l} & -\frac{1}{2}GA \\ \frac{1}{2}GA & -\frac{EJ}{l} + \frac{1}{6}lGA & -\frac{1}{2}GA & \frac{EJ}{l} + \frac{1}{3}lGA \end{bmatrix}$$

Ejemplo: ménsula empotrada en el extremo 1, con carga P en el extremo 2

Primero se resuelve con elementos viga de Timoshenko, con un sólo elemento.

El vector de cargas es

> **carga:=matrix([[V[1]], [M[1]], [P], [0]]);**

$$carga := \begin{bmatrix} V_1 \\ M_1 \\ P \\ 0 \end{bmatrix}$$

Lado izquierdo de la ecuación $K_t \alpha = P$

> **Kta:=multiply(Kt,a);**

$$Kta := \begin{bmatrix} \frac{GAw_1}{l} + \frac{1}{2}GA\theta_1 - \frac{GAw_2}{l} + \frac{1}{2}GA\theta_2 \\ \frac{1}{2}GAw_1 + \left(\frac{EJ}{l} + \frac{1}{3}lGA \right) \theta_1 - \frac{1}{2}GAw_2 + \left(-\frac{EJ}{l} + \frac{1}{6}lGA \right) \theta_2 \\ -\frac{GAw_1}{l} - \frac{1}{2}GA\theta_1 + \frac{GAw_2}{l} - \frac{1}{2}GA\theta_2 \\ \frac{1}{2}GAw_1 + \left(-\frac{EJ}{l} + \frac{1}{6}lGA \right) \theta_1 - \frac{1}{2}GAw_2 + \left(\frac{EJ}{l} + \frac{1}{3}lGA \right) \theta_2 \end{bmatrix}$$

Las ecuaciones que interesan son las dos últimas (las primeras sirven para dar las reacciones)

> **Kta[3,1]=carga[3,1];**

Kta[4,1]=carga[4,1];

$$-\frac{GAw_1}{l} - \frac{1}{2}GA\theta_1 + \frac{GAw_2}{l} - \frac{1}{2}GA\theta_2 = P$$

$$\frac{1}{2}GAw_1 + \left(-\frac{EJ}{l} + \frac{1}{6}lGA \right) \theta_1 - \frac{1}{2}GAw_2 + \left(\frac{EJ}{l} + \frac{1}{3}lGA \right) \theta_2 = 0$$

Sustituyendo las condiciones del empotramiento en 1

> **tmp:=map(z->subs(w[1]=0,theta[1]=0,z),Kta);**

$$tmp := \begin{bmatrix} -\frac{GAw_2}{l} + \frac{1}{2}GA\theta_2 \\ -\frac{1}{2}GAw_2 + \left(-\frac{EJ}{l} + \frac{1}{6}lGA\right)\theta_2 \\ \frac{GAw_2}{l} - \frac{1}{2}GA\theta_2 \\ -\frac{1}{2}GAw_2 + \left(\frac{EJ}{l} + \frac{1}{3}lGA\right)\theta_2 \end{bmatrix}$$

Extraemos los coeficientes que dan la matriz de rigidez efectiva para los 2 g.d.l. del problema:

> **Ke:=matrix([[coeff(Kta[3,1],w[2]),coeff(Kta[3,1],theta[2])],
[coeff(Kta[4,1],w[2]),coeff(Kta[4,1],theta[2])]]);**

$$Ke := \begin{bmatrix} \frac{GA}{l} & -\frac{1}{2}GA \\ -\frac{1}{2}GA & \frac{EJ}{l} + \frac{1}{3}lGA \end{bmatrix}$$

La inversa es la flexibilidad

> **Fe:=inverse(Ke);**

$$Fe := \begin{bmatrix} 4 \frac{(3EJ + l^2 GA)l}{GA(12EJ + l^2 GA)} & 6 \frac{l^2}{12EJ + l^2 GA} \\ 6 \frac{l^2}{12EJ + l^2 GA} & 12 \frac{l}{12EJ + l^2 GA} \end{bmatrix}$$

Veamos la expresión que resulta si se elimina el siguiente factor común de la matriz:

$$tmp1 := \frac{1 + \frac{12EJ}{GA l^2}}{\frac{12EJ}{GA l^2}}$$

>

A falta del factor $tmp1$, la matriz de flexibilidad es:

> **Fe1:=map(expand,map(simplify,evalm(tmp1*Fe)));**

$$F_{el} := \begin{bmatrix} \frac{l}{GA} + \frac{1}{3} \frac{l^3}{JE} & \frac{1}{2} \frac{l^2}{JE} \\ \frac{1}{2} \frac{l^2}{JE} & \frac{l}{JE} \end{bmatrix}$$

Calculemos ahora el valor exacto. Para ello, repetimos lo mismo pero con elementos viga de Bernouilli (exactos para flexión)

> **Kba:=map(z->subs(w[1]=0,theta[1]=0,z),multiply(Kb,a));**

$$Kba := \begin{bmatrix} -12 \frac{EJw_2}{l^3} + 6 \frac{EJ\theta_2}{l^2} & \\ \\ -6 \frac{EJw_2}{l^2} + 2 \frac{EJ\theta_2}{l} & \\ \\ 12 \frac{EJw_2}{l^3} - 6 \frac{EJ\theta_2}{l^2} & \\ \\ -6 \frac{EJw_2}{l^2} + 4 \frac{EJ\theta_2}{l} & \end{bmatrix}$$

> **Ke:=matrix([[coeff(Kba[3,1],w[2]),coeff(Kba[3,1],theta[2])], [coeff(Kba[4,1],w[2]),coeff(Kba[4,1],theta[2])]]);**

$$K_e := \begin{bmatrix} 12 \frac{EJ}{l^3} & -6 \frac{EJ}{l^2} \\ -6 \frac{EJ}{l^2} & 4 \frac{EJ}{l} \end{bmatrix}$$

> **Feb:=inverse(Ke);**

$$F_{eb} := \begin{bmatrix} \frac{1}{3} \frac{l^3}{JE} & \frac{1}{2} \frac{l^2}{JE} \\ \frac{1}{2} \frac{l^2}{JE} & \frac{l}{JE} \end{bmatrix}$$

Añadimos la flecha por cortante y tenemos la flexibilidad de la solución exacta:

$$Feb_{1,1} := Feb_{1,1} + \frac{l}{GA}$$

> **evalm(Feb);**

$$\begin{bmatrix} \frac{l}{GA} + \frac{1}{3} \frac{l^3}{JE} & \frac{1}{2} \frac{l^2}{JE} \\ \frac{1}{2} \frac{l^2}{JE} & \frac{l}{JE} \end{bmatrix}$$

> **evalm(Fe1)=evalm(Feb);**

$$\begin{bmatrix} \frac{l}{GA} + \frac{1}{3} \frac{l^3}{JE} & \frac{1}{2} \frac{l^2}{JE} \\ \frac{1}{2} \frac{l^2}{JE} & \frac{l}{JE} \end{bmatrix} = \begin{bmatrix} \frac{l}{GA} + \frac{1}{3} \frac{l^3}{JE} & \frac{1}{2} \frac{l^2}{JE} \\ \frac{1}{2} \frac{l^2}{JE} & \frac{l}{JE} \end{bmatrix}$$

Vemos que se obtiene idéntico resultado que antes, pero en este caso no aparece el factor **tmp1**.
 Veamos el valor de este factor multiplicador, que indica la rigidización artificial de la ménsula, para un caso concreto con sección rectangular de ancho b , canto t y módulo de Poisson $\nu = \frac{1}{4}$.


$$\text{tmp2}(\lambda) := \text{subs} \left(\left\{ J = \frac{bt^3}{12}, A = \frac{5bt}{6}, G = \frac{E}{2(1+\nu)}, l = t\lambda \right\}, \text{tmp1} \right)$$

$$\text{tmp2}(\lambda) := \frac{5}{6} \frac{\left(1 + \frac{6(2+2\nu)}{5\lambda^2} \right) \lambda^2}{2+2\nu}$$

> **tmp2(lambda):=simplify(subs(nu=1/4,tmp2(lambda)));**

$$\text{tmp2}(\lambda) := \frac{1}{3} \lambda^2 + 1$$

> **plot(tmp2(lambda),lambda=1..50);**



Viga de Timoshenko con integración reducida

Integrando de la rigidez a cortante

> **Kc1:=multiply(transpose(Bc),(G*A)*Bc);**

$$Kc1 := \begin{bmatrix} \frac{GA}{l^2} & -\frac{GA\left(\frac{x}{l}-1\right)}{l} & -\frac{GA}{l^2} & \frac{GAx}{l^2} \\ -\frac{GA\left(\frac{x}{l}-1\right)}{l} & GA\left(\frac{x}{l}-1\right)^2 & \frac{GA\left(\frac{x}{l}-1\right)}{l} & -\frac{\left(\frac{x}{l}-1\right)GAx}{l} \\ -\frac{GA}{l^2} & \frac{GA\left(\frac{x}{l}-1\right)}{l} & \frac{GA}{l^2} & -\frac{GAx}{l^2} \\ \frac{GAx}{l^2} & -\frac{\left(\frac{x}{l}-1\right)GAx}{l} & -\frac{GAx}{l^2} & \frac{x^2 GA}{l^2} \end{bmatrix}$$

Particularizamos en el único punto de integración que emplearemos, en el centro del elemento ($x = \frac{l}{2}$)

> **Kc1:=map(z->subs(x=l/2,z),Kc1);**

$$K_{cl} := \begin{bmatrix} \frac{GA}{l^2} & \frac{1}{2} \frac{GA}{l} & -\frac{GA}{l^2} & \frac{1}{2} \frac{GA}{l} \\ \frac{1}{2} \frac{GA}{l} & \frac{1}{4} GA & -\frac{1}{2} \frac{GA}{l} & \frac{1}{4} GA \\ -\frac{GA}{l^2} & -\frac{1}{2} \frac{GA}{l} & \frac{GA}{l^2} & -\frac{1}{2} \frac{GA}{l} \\ \frac{1}{2} \frac{GA}{l} & \frac{1}{4} GA & -\frac{1}{2} \frac{GA}{l} & \frac{1}{4} GA \end{bmatrix}$$

Integrando:

> **Kc1:=map(z->int(z,x=0..l),Kc1);**

$$K_{cl} := \begin{bmatrix} \frac{GA}{l} & \frac{1}{2} GA & -\frac{GA}{l} & \frac{1}{2} GA \\ \frac{1}{2} GA & \frac{1}{4} l GA & -\frac{1}{2} GA & \frac{1}{4} l GA \\ -\frac{GA}{l} & -\frac{1}{2} GA & \frac{GA}{l} & -\frac{1}{2} GA \\ \frac{1}{2} GA & \frac{1}{4} l GA & -\frac{1}{2} GA & \frac{1}{4} l GA \end{bmatrix}$$

Comparación entre la integración completa (K_c) y la reducida de 1 punto (K_{cl})

> **evalm(Kc)=evalm(Kc1);**

$$\begin{bmatrix} \frac{GA}{l} & \frac{1}{2} GA & -\frac{GA}{l} & \frac{1}{2} GA \\ \frac{1}{2} GA & \frac{1}{3} l GA & -\frac{1}{2} GA & \frac{1}{6} l GA \\ -\frac{GA}{l} & -\frac{1}{2} GA & \frac{GA}{l} & -\frac{1}{2} GA \\ \frac{1}{2} GA & \frac{1}{6} l GA & -\frac{1}{2} GA & \frac{1}{3} l GA \end{bmatrix} = \begin{bmatrix} \frac{GA}{l} & \frac{1}{2} GA & -\frac{GA}{l} & \frac{1}{2} GA \\ \frac{1}{2} GA & \frac{1}{4} l GA & -\frac{1}{2} GA & \frac{1}{4} l GA \\ -\frac{GA}{l} & -\frac{1}{2} GA & \frac{GA}{l} & -\frac{1}{2} GA \\ \frac{1}{2} GA & \frac{1}{4} l GA & -\frac{1}{2} GA & \frac{1}{4} l GA \end{bmatrix}$$

> **Kt1:=Kf+Kc1;**

$$K_{t1} := K_f + K_{cl}$$

Repetimos el ejemplo anterior de la ménsula:

> **Kta:=multiply(Kt1,a);**

$$Kta := \begin{bmatrix} \frac{GAw_1}{l} + \frac{1}{2}GA\theta_1 - \frac{GAw_2}{l} + \frac{1}{2}GA\theta_2 \\ \frac{1}{2}GAw_1 + \left(\frac{EJ}{l} + \frac{1}{4}lGA\right)\theta_1 - \frac{1}{2}GAw_2 + \left(-\frac{EJ}{l} + \frac{1}{4}lGA\right)\theta_2 \\ -\frac{GAw_1}{l} - \frac{1}{2}GA\theta_1 + \frac{GAw_2}{l} - \frac{1}{2}GA\theta_2 \\ \frac{1}{2}GAw_1 + \left(-\frac{EJ}{l} + \frac{1}{4}lGA\right)\theta_1 - \frac{1}{2}GAw_2 + \left(\frac{EJ}{l} + \frac{1}{4}lGA\right)\theta_2 \end{bmatrix}$$

Las ecuaciones que interesan son las dos últimas (las primeras sirven para dar las reacciones)

> **Kta[3,1]=carga[3,1];**

Kta[4,1]=carga[4,1];

$$-\frac{GAw_1}{l} - \frac{1}{2}GA\theta_1 + \frac{GAw_2}{l} - \frac{1}{2}GA\theta_2 = P$$

$$\frac{1}{2}GAw_1 + \left(-\frac{EJ}{l} + \frac{1}{4}lGA\right)\theta_1 - \frac{1}{2}GAw_2 + \left(\frac{EJ}{l} + \frac{1}{4}lGA\right)\theta_2 = 0$$

Sustituyendo las condiciones del empotramiento en 1

> **tmp:=map(z->subs(w[1]=0,theta[1]=0,z),Kta);**

$$tmp := \begin{bmatrix} -\frac{GAw_2}{l} + \frac{1}{2}GA\theta_2 \\ -\frac{1}{2}GAw_2 + \left(-\frac{EJ}{l} + \frac{1}{4}lGA\right)\theta_2 \\ \frac{GAw_2}{l} - \frac{1}{2}GA\theta_2 \\ -\frac{1}{2}GAw_2 + \left(\frac{EJ}{l} + \frac{1}{4}lGA\right)\theta_2 \end{bmatrix}$$

Extraemos los coeficientes que dan la matriz de rigidez efectiva para los 2 g.d.l. del problema:

> **Ke:=matrix([[coeff(Kta[3,1],w[2]),coeff(Kta[3,1],theta[2])],
[coeff(Kta[4,1],w[2]),coeff(Kta[4,1],theta[2])]]);**

$$K_e := \begin{bmatrix} \frac{GA}{l} & -\frac{1}{2}GA \\ -\frac{1}{2}GA & \frac{EJ}{l} + \frac{1}{4}lGA \end{bmatrix}$$

La inversa es la flexibilidad

> **Fe:=map(expand,inverse(Ke));**

$$F_e := \begin{bmatrix} \frac{l}{GA} + \frac{1}{4} \frac{l^3}{JE} & \frac{1}{2} \frac{l^2}{JE} \\ \frac{1}{2} \frac{l^2}{JE} & \frac{l}{JE} \end{bmatrix}$$

Comparación con la flexibilidad exacta: ya no aparece el término responsable del bloqueo, aunque la solución difiere en el primer término (de $\frac{1}{3}$ a $\frac{1}{4}$). Esto origina un error que a medida que la

discretización es más fina, se reduce. Si empleásemos un sólo elemento para la ménsula, en el límite de esbeltez infinita, la flecha calculada sería $\frac{3}{4}$ de la exacta. (Como hemos visto, esto es mucho

más dramático sin integración reducida: ¡la ménsula se bloquea y la flecha sale nula!)

> **evalm(Feb)=evalm(Fe);**

$$\begin{bmatrix} \frac{l}{GA} + \frac{1}{3} \frac{l^3}{JE} & \frac{1}{2} \frac{l^2}{JE} \\ \frac{1}{2} \frac{l^2}{JE} & \frac{l}{JE} \end{bmatrix} = \begin{bmatrix} \frac{l}{GA} + \frac{1}{4} \frac{l^3}{JE} & \frac{1}{2} \frac{l^2}{JE} \\ \frac{1}{2} \frac{l^2}{JE} & \frac{l}{JE} \end{bmatrix}$$

>